Escola Politécnica da Universidade de São Paulo

Lista E

PME 3380 - Modelagem de Sistemas Dinâmicos

Integrante:
Bruno Akira Oshiro NUSP: 10771667

Exercício:



Primeiramente, calcula-se a equação diferencial que rege o movimento. Pelo TMB, sabemos que:

$$m\ddot{x} = F(t) - c\dot{x} - kx$$

E na forma de espaço de estados:

$$\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix} = \begin{bmatrix} 0 & 1 \\ -k/m & -c/m \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} + \begin{bmatrix} 0 \\ 1/m \end{bmatrix} F(t)$$

Dessa forma, temos o seguinte sistema, com F(t)=u:

$$\begin{cases} \dot{x}_1 = x_2 \\ \dot{x}_2 = -\dfrac{k}{m}x_1 - \dfrac{c}{m}x_2 + \dfrac{1}{m}u \end{cases}$$

Realizando a transformada de Laplace:

$$\begin{cases} sX_1 - x(0) = X_2 \\ sX_2 - x(0) = -\dfrac{k}{m}X_1 - \dfrac{c}{m}X_2 + \dfrac{1}{m}U \end{cases}$$

Como estamos interessados no estudo do regime particular do sistema, podemos zerar as condições iniciais:

$$\begin{cases} sX_1 = X_2 \\ sX_2 = -\dfrac{k}{m}X_1 - \dfrac{c}{m}X_2 + \dfrac{1}{m}U \end{cases}$$

A partir do sistema adquirido, podemos calcular a função de transferência G(s):

$$G(s) = \frac{X_1(s)}{U(s)} = \frac{1}{ms^2 + cs + k}$$

Simulou-se o sistema para diferentes valores de ζ:

- $\boldsymbol{m = 1; c = 1; k = 81; \zeta < 1}$**:** Observa-se que há uma oscilação em torno do ponto de equilíbrio $kx_{eq} = F(t) \rightarrow x_{eq} = F(t)/k \ = \ 0.012\, m$.



- $\boldsymbol{m = 1; c = 18; k = 81; \zeta = 1}$**:** Observa-se que não há mais oscilação até o ponto de equilíbrio $x_{eq} = F(t)/k \ = \ 0.012\, m$.



- $\boldsymbol{m = 1; c = 30; k = 81; \zeta > 1}$**:** Observa-se que, novamente, não há mais oscilação até o ponto de equilíbrio $x_{eq} = F(t)/k \ = \ 0.012m$. Nesse caso, o tempo para a deflexão atingir o equilíbrio é maior que no caso anterior, por conta do aumento de ζ.



Lição de casa:

1)

- **Cálculo dos autovalores da matriz A:**

$$\det\begin{vmatrix} 0-\lambda & 1 \\ -k/m & -\dfrac{c}{m}-\lambda \end{vmatrix} = 0$$

$$m\lambda^2 + c\lambda + k = 0$$

$$\lambda = \frac{-c \pm \sqrt{c^2 - 4mk}}{2m}$$

- **Cálculo das raízes do polinômio no denominador da função de transferência:**

$$ms^2 + cs + k = 0$$

$$s = \frac{-c \pm \sqrt{c^2 - 4mk}}{2m}$$

Observa-se que os resultados dos autovalores e as raízes do polinômio são exatamente o mesmo.

- **Análise para $\zeta < 1$, ou seja:**

$$\frac{c}{2\sqrt{km}} < 1 \Rightarrow c^2 < 4mk$$

$$\therefore c^2 - 4mk < 0$$

Dessa forma, o que está dentro da raiz quadrada é negativa, assim, o valor de s é um número imaginário.

- **Cálculo e comparação do módulo de s com ω_n:**

$$s = \frac{1}{2m}\left(-c + \sqrt{|c^2 - 4mk|}\; i\right)$$

$$s = \frac{1}{2m}\left(-c + \sqrt{4mk - c^2}\; i\right)$$

$$|s| = \sqrt{\frac{1}{4m^2}(c^2 + 4mk - c^2)} = \sqrt{\frac{k}{m}} \quad \Rightarrow \quad \therefore |s| = \omega_n$$

- **Coeficiente de amortecimento:**

$$\frac{\left|-\frac{c}{2m}\right|}{\sqrt{\frac{k}{m}}} = \frac{c}{2m}\sqrt{\frac{m}{k}} = \frac{c}{2\sqrt{mk}} = \zeta$$

- **Observe que a frequência de oscilação é igual ao módulo da parte imaginária do pólo:**

$$parte\ imagin\acute{a}ria = \frac{\sqrt{4mk - c^2}}{2m} = \sqrt{\frac{k}{m}} * \sqrt{1 - \frac{c^2}{4mk}} = \omega_n\sqrt{1 - \zeta^2} \;\; \therefore \omega_d$$

2) Plotou-se a solução do exercício anterior para diferentes casos sem forças externas. Em todos eles, fixou-se os parâmetros do sistema e as condições iniciais foram variadas de:

Condições	$x(0)$	$\dot{x}(0)$
I	1	0
II	2	0
III	3	0
IV	4	0

- **Duas raízes complexas.** $\zeta < 1; m = 1; c = 1; k = 81$:





- **Duas raízes reais iguais.** $\zeta = 1; m = 1; c = 18; k = 81$:





- **Duas raízes reais distintas.** $\zeta > 1; m = 1; c = 30; k = 81$:





Códigos utilizados

Para a parte do **exercício**, utilizou-se o seguinte programa:

```
// Definindo os parametros do sistema:
m=1;c=1;k=81;
// Matrizes do sistema:
A=[0 1; -k/m -c/m];
B=[0;1/m];
C=[1 0];
D=[0];
// Montando o sistema:
suspensao=syslin('c',A,B,C,D);
// Definindo o vetor tempo:
t=0:0.01:2;
// Definindo a entrada:
u=ones(t);
// No espaco de estados temos 2 variaveis de estado:
x0e=[0;0]; // neste caso, x1(0)=0 e x2(0)=0
// Alem de calcular a saida y, a função csim também permite obter o estado x:
[y,x]=csim(u,t,suspensao,x0e);
// Abrindo uma nova janela de graficos:
xset('window',1)
// Mostrando o resultado da simulacao:
plot2d(t,y,2)
xtitle('Resposta a degrau','tempo t','Deformacao da mola')
// Podemos plotar o grafico do estado x2, por exemplo:
// Abrindo uma nova janela de graficos:
xset('window',2)
// Mostrando o resultado da simulacao:
plot2d(t,x(2,:),2)
xtitle('Resposta a degrau','tempo t','Velocidade da massa')
```

Para a parte da **lição de casa**, utilizou-se o seguinte programa:

```
clear();

function [x]=Solucao(x0e, i, m, c, k, t)
// Matrizes do sistema:
A=[0 1; -k/m -c/m];
B=[0;1/m];
C=[1 0];
D=[0];
// Montando o sistema:
suspensao=syslin('c',A,B,C,D);
// Definindo a entrada degrau:
u=zeros(t);
// No espaco de estados temos 2 variaveis de estado:
//x0e=[0;0]; // neste caso, x1(0)=0 e x2(0)=0
// Alem de calcular a saida y, a função csim também permite obter o estado x:
[y,x]=csim(u,t,suspensao,x0e);

endfunction

// Definindo os parametros do sistema:
m=1;c=1;k=81;
// Definindo o vetor tempo:
t=0:0.01:2;
//condições iniciais que deseja analisar
x0=[1 2 3 4];
x0_linha=[0 0 0 0];
//resgatando o tamanho do vetor
qtd_cond=length(x0);
//vetores de resultados
```

```
X=zeros(qtd_cond,length(t));
V=zeros(qtd_cond,length(t));

for i=1:qtd_cond
   x0e=[x0(i);x0_linha(i)];
   [x]=Solucao(x0e,i,m,c,k,t)
   X(i,:)=x(1,:); //deflexão da mola
   V(i,:)=x(2,:); //valocidade da massa

end

// Abrindo uma nova janela de graficos:
xset('window',1)
// Plotando o resultados:
plot(V(1,:),X(1,:),"k",V(2,:),X(2,:),"r",V(3,:),X(3,:),"b",V(4,:),X(4,:),"g");
//colocando legendas
legend(["Condição I";"Condição II";"Condição III";"Condição IV"]);
//aumentando as fontes
b=gce();
b.font_size=3
xtitle("Gráfico de velocidade x deflexão","Deflexão","Velocidade");
a=gca();
fonte=3
a.font_size=fonte
a.x_label.font_size=fonte
a.y_label.font_size=fonte
a.title.font_size=fonte

//encontrando as raízes do polinomio
s1=poly([k c m],"s",'c')
solution=roots(s1)
//plotando as raizes
xset('window',2)
plot(real(solution(1)),imag(solution(1)),'o',real(solution(2)),imag(solution(2)),'o','linewidth',8)

xtitle("Pólos","Re","Im");
```